Errata

Document Title: Constant Current Measurements Using the 4194A (AN 339-8)

Part Number: 5950-2923

Revision Date: December 1986

HP References in this Application Note

This application note may contain references to HP or Hewlett-Packard. Please note that Hewlett-Packard's former test and measurement, semiconductor products and chemical analysis businesses are now part of Agilent Technologies. We have made no changes to this application note copy. The HP XXXX referred to in this document is now the Agilent XXXX. For example, model number HP8648A is now model number Agilent 8648A.

About this Application Note

We've added this application note to the Agilent website in an effort to help you support your product. This manual provides the best information we could find. It may be incomplete or contain dated information, and the scan quality may not be ideal. If we find a better copy in the future, we will add it to the Agilent website.

Support for Your Product

Agilent no longer sells or supports this product. You will find any other available product information on the Agilent website:

www.agilent.com

Search for the model number of this product, and the resulting product page will guide you to any available information. Our service centers may be able to perform calibration if no repair parts are needed, but no other support from Agilent is available.

Agilent Technologies

HEWLETT PACKARD

Application Note 339-8

Constant Current Measurements Using the HP 4194A

– HP 4194A Impedance/Gain–Phase Analyzer –



Introduction

The HP 4194A Impedance/Gain-Phase Analyzer, with its advanced impedance measurement capabilities up to 40MHz (up to 100MHz when using the HP 41941A/B Impedance Probe Kit), high resolution internal color CRT, and equivalent circuit analysis function, is ideally suited to the evaluation of inductors and crystal resonators. You can also use the HP 4194A for constant current applications, even though the HP 4194A's oscillator level provides a constant voltage, by using the Auto Sequence Program (ASP) internal programming function. Because the HP 4194A has a level monitor function, constant signal currents can be easily programmed. The variable oscillator voltage of 10mV to 1V (10mV to 1.26V when using the HP 41941A/B, Option 350) makes it possible to provide constant currents, thereby improving evaluation efficiency for current-dependent devices at lower test cost. This application note will be helpful in inductor manufacturers (e.g., coils, transformers, and magnetic heads),inductor users (e.g., regulator manufacturers),crystal resonator manufacturers, and crystal resonator users.

Measurement Requirements and HP 4194A Solutions

Many inductive devices, such as cored inductors and transformers, and high dielectric devices, such as crystal resonators, are current dependent, and should be tested under constant current conditions. Since most LCR meters provide a constant voltage oscillator level, an external computer is required to provide the programming necessary for constant current applications. Until now, that is.

By using the HP 4194A's ASP and test current monitor functions, constant current measurements can be programmed with ease. Provided here are two ASP programs, one for performing fast constant current measurements (without over-current protection) where current accuracy is not vitally important, for testing such devices as crystal resonators, and another ASP program for performing highly accurate constant current measurements.

ASP Programs

High Speed version
(with no over-current protection)

Use this program when:

* Measurement speed is preferred over test current accuracy.
* Testing devices whose impedance charac teristics are not overly test signal dependent.

When controlling the oscillator voltage to obtain a constant current, the following equation is used to determine the correct voltage setting.

$$Vc = Icon \times \frac{Vi}{Imon}$$

where Vc is the correct oscillator voltage setting
Icon is the specified constant current
Vi is the initial oscillator voltage setting
Imon is the monitored current when Vi is applied



Figure 1 High-speed version flow chart

Before you begin making measurements, set up an initial oscillator voltage setting and measure the test current to estimate the correct oscillator voltage setting. Once the correct oscillator voltage is established, you can perform constant current measurements at the desired oscillator voltage settings. Figure 1 shows the flow chart and Figure 2 shows the program list for this high speed version ASP program.

Note:

* When measuring such test devices as crystal resonators, the test current must not exceed a specified value or the resonator's characteristics may change radically because of its high dielectric constant. Set Vi to a low value to avoid excessive current.
* Test current accuracy for this method depends on the linearity of the test device. The better the device's linearity, the more accurate the test current will be.

```
 10 !I CONST (HIGH SPEED)
 20 RST
 30 CMT"";SWT2;IVM2;SWP1;ASC2
 40 START=100 KHZ
 50 STOP=40 MHZ
 60 R1=21              ! NO.OF POINT
 70 NOP=R1
 80 R11=.005
 90 R10=100U           ! I CON.
100 R99=1;R90=.01      ! OSC LIMIT
110 !
120 OSC=.07            ! INITIAL OSC
130 TRGM2
140 !
150 ITM1
160 FOR R0=1 TO R1
170  SWM3;MANUAL=X(R0)
180  TRIG
190  RA(R0)=R10*OSC/MON
200 NEXT R0
210 !
220 ITM1
230 FOR R0=1 TO R1
240  IF RA(R0)>R99 THEN RA(R0)=R99;GOTO 260
250  IF RA(R0)<R90 THEN RA(R0)=R90
260  OSC=RA(R0)
270  WAIT 100
280  SWM3;MANUAL=X(R0)
290  TRIG
300  RB(R0)=MON
310 NEXT R0
320 AUTOA
330 BEEP
340 DISP "PRESS CONT"
350 PAUSE
360 CMT"*******************   OSC(V)   MONITOR I(A)"
370 DSP3;UNIT0
380 A=RA;B=RB
390 BEEP
400 DISP "COPY TO PRINTER?  Y=1/N=0"
410 PAUSE
420 IF Z=0 THEN GOTO 470
430 CMT"A= OSC(V)  B= MONITOR I(A)"
440 CPYM2
450 COPY
460 CPYM3
470 END
```

R0 : FOR NEXT counter for frequency sweep
R10: Constant current (Icon)
R1 : Number of points
R11: Maximum OSC level resolution
R90: Minimum HP 4194A OSC level
R99: Maximum HP 4194A OSC level
RA(XX): OSC level (Vi)
RB(XX): Monitor current (Imon)

Figure 2. High-speed version ASP program

High accuracy version

Use this program when:
* Constant current accuracy is preferred over measurement speed.

In this method, a linear interpolation search is performed to determine the oscillator voltage necessary to obtain the required constant current. The oscillator voltage setting is derived from the difference value between the V1/I1 and V2/I2 points as shown in Figure 3. Maximum constant current accuracy is achieved when the maximum oscillator level setting resolution of 10mV is reached. Figure 4 shows the flow chart and Figure 5 shows the ASP program listing . Figures 6 and 7 show constant current measurement results obtained when using the high accuracy version ASP program.

Figure 3. Linear interpolation model



V3: Result of a linear interpolation cycle.

Figure 4. High-accuracy version flow chart



Figure 5. High-accuracy version ASP program

```
10 ! I CONST (HIGH ACCURACY)
20 RST
30 CMT"";FNC1;IMP1;SWT2;IVM2;SWP1;ASC2
40 START=100 KHZ
50 STOP=40 MHZ
60 R1= 51              ! NO.OF POINT
70 NOP=R1
80 R11=.001
90 R10=100U            ! I CON.
100 R99=1;R90=.01      ! OSC LIMIT
110 !
120 OSC=.05            ! INITIAL OSC
130 TRGM2
140 !
150 FOR R0=1 TO R1
160  SWM3;MANUAL=X(R0)
170  TRIG
180  R20=OSC/MON
190  RA(R0)=R10*R20
200  IF RA(R0)>R99 THEN RA(R0)=R99 ;GOTO 220
210  IF RA(R0)<R90 THEN RA(R0)=R90
220  OSC=RA(R0)
230  WAIT 100
240  TRIG
250  RB(R0)=MON
260  IF RB(R0)>R10 THEN R21=R90;R22=RA(R0);R31=R90/R20;R32=RB(R0);GOTO 280
270                     R21=RA(R0);R22=R99;R31=RB(R0);R32=R99/R20
280   R23=RA(R0)
290   IF R32=R31 THEN R24=R23 ;GOTO 430
300   R23=(R22-R21)*(R10-R31)/(R32-R31)+R21
310   IF ABS(RA(R0)-R23)<R11 THEN 430
320   RA(R0)=R23
330   IF R23>R99 THEN R23=R99;GOTO 350
340   IF R23<R90 THEN R23=R90
350   OSC=R23
360   WAIT 100
370   SWM3;MANUAL=X(R0)
380   TRIG
390   RB(R0)=MON
400   IF RB(R0)<=R10 THEN R21=R23;R31=RB(R0);GOTO 420
410                      R22=R23;R32=RB(R0)
420   IF ABS(RB(R0)/R10-1)>.0001 THEN 290
430  !
440 NEXT R0
450 AUTOA
460 BEEP
470 DISP "PRESS CONT"
480 PAUSE
490 CMT"********************   OSC(V)   MONITOR I(A)"
500 DSP3;UNIT0
510 A=RA;B=RB
520 BEEP
530 DISP "COPY TO PRINTER?  Y=1/N=0"
540 PAUSE
550 IF Z=0 THEN GOTO 600
560 CMT"A= OSC(V)  B= MONITOR I(A)"
570 CPYM2
580 COPY
590 CPYM3
600 END
```

R0 : FOR NEXT counter for frequency sweep
R10: Constant current (Icon)
R1 : Number of points
R11: Maximum OSC level resolution
R90: Minimum HP 4194A OSC level
R99: Maximum HP 4194A OSC level
RA(XX): OSC level (Vi)
RB(XX): Monitor current (Imon)

R21: V1
R22: V2
R31: I1
R32: I2
R23: V3

Figure 6. Constant current measurements of an inductor (51 point measurement)



Figure 7. Constant current measurements of an inductor (21 point measurement)

A= OSC(V) B= MONITOR I(A)

N	FREQUENCY [Hz]	[ ]		[ ]	
1	100 000.000	10.0000	m	162.210	u *
2	134 928.285	10.0000	m	143.119	u *
3	182 056.420	10.0000	m	122.365	u *
4	245 645.605	10.0000	m	100.409	u
5	331 445.402	12.5310	m	99.7758	u
6	447 213.595	16.0774	m	100.269	u
7	603 417.634	21.1689	m	99.9757	u
8	814 181.063	28.7875	m	100.107	u
9	1 098 560.543	39.7208	m	100.333	u
10	1 482 268.898	54.7709	m	99.9111	u
11	2 000 000.000	82.5323	m	99.7481	u
12	2 698 565.695	145.956	m	100.332	u
13	3 641 128.406	416.531	m	100.038	u
14	4 912 912.104	543.954	m	100.059	u
15	6 628 908.035	166.047	m	99.7903	u
16	8 944 271.910	92.8976	m	100.004	u
17	12 068 352.673	61.9393	m	100.134	u
18	16 283 621.261	43.9394	m	99.5811	u
19	21 971 210.866	30.6516	m	100.217	u
20	29 645 377.964	28.8073	m	99.8141	u
21	40 000 000.000	20.9204	m	99.7913	u

* The HP 4194A's OSC level has been reached minimum level.

Printed in Japan December 1986 5950-2923